
ABSTRACT The SAS R© include statement is simple, yet powerful. This paper reviews Fehd [3,
sgf2007.028] FreqAll data review program which produces a shortened data review report
of frequencies of each variable in a data set. It provides routines called using call execute
and %include to produce the same report for every data set in a libref.

INTRODUCTION

List of Topics These are the topics discussed in the introduction.

Topic Page

Compare Algorithms 2
Example Output 3
Review of Concepts 4
Review of Main Module A0Smry 5

Overview This paper examines the list processing issues of calling one routine many times, using
data set variables as a list of parameters. FreqAll uses SQL to generate macro calls to its
subroutine FreqOf. FreqLibname uses call execute to generate calls to its routines which
are parameterized include files.

COMPARE ALGORITHMS

FreqAll Fehd’s FreqAll consisted of two parts: macro FreqOf and the FreqAll list processing routine.
The calls to the subroutine macro FreqOf were generated by SQL writing text into a macro
variable. The limitation of this algorithm is the program may run out of memory for the macro
variable symbol table.
Example macro calls, which were all in one macro variable:

1 %FreqOf(Libname = sashelp
2 ,Memname = Class
3 ,Name = Height
4 ,Type = N
5 )

FreqLibname The macro FreqAll.FreqOf has been replaced with a parameterized include file, ProcFreq.
The repetition of calls is handled by call execute.
Example parameterized include file calls:

1 %Let Libname = sashelp;
2 %Let Memname = Class ;
3 %Let Name = Height ;
4 %Let Type = N ;
5 %Include Project(ProcFreq);

FreqLibname uses Proc Contents to make the lists of variables and list of data sets. These
lists are used by each of the routines, CallXProc, and CallXRpt, to call execute the %include
of the subroutines: ProcFreq, ProcSmry, RptMemName.

Map of Calls These tables show the calls of routines and subroutines by each module.

FreqAll
Map of Routine and Subroutine Calls
main routine subroutines
FreqAll MakLists

SQL FreqOf
RptMemN

FreqLibname
Map of Routine and Subroutine Calls

main routines routines subroutines
A0Smry MakLists

CallXProc ProcFreq DatStruc
ProcMode

ProcSmry DatStruc
CallXRpt RptMemN

EXAMPLE OUTPUT

Overview The FreqLibname Report, per Memname, is a listing which contains:

attributes data structure list similar to Proc Contents

summary abbreviated freqency showing the high and low values and the number of levels;
for numeric variables: mode, min, mean, max, n and nmiss

Attributes The data structure is the primary item for consideration in data review.
There are two other considerations for each variable:

1. how many levels does each variable have?

2. is the variable unique, i.e. does the number of levels equal the number of observa-
tions?

The FreqLibname listing contains the data set name in the title with the number of observa-
tions. As a data manager I am concerned to discover whether the data set has both labels
and formats. If they are missing, I have to provide them.
Example report of data set attributes, see the demonstration file zqCITIDAYreport.txt:

1 SmryLibname
2 Report Memname: SASHELP.CITIDAY nobs=1069 nvars=11
3 Report Memname: MemLabel=Citibase daily indicators: JAN88 FEB92
4 Report Memname: attributes
5

6 Var
7 Num Name Type Length Label Format Level Unique
8

9 1 DATE N 7 Date of Observation DATE9. 1069 1
10 10 DCD1M N 8 INT.RATE:1MO CERTIFICATE 388 0
11 9 DCP07 N 8 7 DAY COMMERCIAL PAPER 324 0
12 ...

Summary The FreqAll routine provided only the output from Proc Freq. FreqLibname provides addi-
tional information:

1. when a format is present, the formatted value

2. number of levels, and a note if the variable is unique

3. Proc Summary information; note: other Proc Summary information can be added

Report Titles

1 SmryLibname
2 Report Memname: SASHELP.CITIDAY nobs=1069 nvars=11
3 Report Memname: MemLabel=Citibase daily indicators: JAN88:FEB92
4 Report Memname: summary
5

6 Type
7 Name Len Valu C formatted Valu N N % Level

Proc Freq Note: this example shows only the two lowest and highest values. The ProcFreq
subroutine contains a parameter, Nobs2View, which controls how many levels to show.

1 DATE N.7 . 01JAN1988 10227.00 1 0.09 1
2 . 04JAN1988 10230.00 1 0.09 2
3 . 04FEB1992 11722.00 1 0.09 1068
4 . 05FEB1992 11723.00 1 0.09 1069
5 levels=1069:is.primekey? . . . . 1069

ProcMode There is not a Proc Mode; see the ProcMode subroutine.

1 mode 01JAN1988 10227.00 1 0.09 2
2 mode 01JAN1988 10230.00 1 0.09 3
3 mode 01JAN1988 10231.00 1 0.09 4

Proc Summary Note: other statistics may be added.

1 min . 10227.00 . . .
2 mean . 10975.40 . . .
3 median . 10975.00 . . .
4 max . 11723.00 . . .
5 n . 1069.00 . 100.00 .
6 nmiss . 0.00 . 0.00 .

Note: Percent is calculated for the statistics N and Nmiss.

REVIEW OF CONCEPTS

Using Includes The %Include statement opens and reads all statements in a file. The option source2
controls whether the statements are echoed to the log. The default value is nosource2.
Note that the routines and subroutines check the value of option source2 in order to self
report while testing.

Functions and
Call Routines

These functions are used in the programs.

call execute submits statement for execution in next step; see also %nrstr

cat functions concatenation functions, replaces concatenation operator (!!)

cat: no trim

cats: remove leading and trailing blanks

catt: remove trailing blanks

catx: remove leading and trailing blanks, add separator specified in first argument

%eval evaluate numeric expression, return integer; used to test value of options during
testing; see %sysfunc getoption

link goto named label; code is bracketed by return; statements

%nrstr: No Rescan String forces resolution of macro variable assignments and calls in
next step; used with call execute

putlog write note to log; eliminates use of file log; statement

%sysfunc getoption returns current value of option in all caps

vname returns name of variable as text from array reference

REVIEW OF MAIN MODULE: A0Smry

Overview This section examines the parts of the main program A0Smry.sas.

1. Parameters

2. Processing

3. Optional Reports

Parameters The primary parameters are:

Libname: Libref of reports

1 ** 1 Prepare SmryLibname report for:;
2 %Let Libname = Library;

LibWork: Libref of list processing data sets

1 ** 2 Store Smry* data sets in libref:;
2 %Let LibWork = Work;

Path2Txt: output file prefix, may include folder name

1 ** 3 Write Smry report *.txt to folder:;
2 %Let Path2Txt = zq;%*here: zq&MemName.*.txt;

Processing Input Program MakLists creates a data set which is used as a list of parameters for routines.

1 *input : Make lists for CallX*;
2 %Include Project(MakLists) ;

Process Program CallXproc calls the subroutines ProcFreq, ProcMode, and ProcSmry

1 *process: Call procs freq, mode and summary;
2 %Include Project(CallXProc) ;

Output Program CallXrpt calls the reporting subroutine

1 *output : Print summary report, by MemName ;
2 %Include Project(CallXRpt) ;

Optional
Reports

Additional programs are provided in the suite .zip for the following tasks:

RptNameA: by variable Name; compare that same named variables in different data sets
have the same attributes: Type, Length and Label

WriteAttrib: write an attribute statement for the data set; if length needs changing or for-
mats or labels are missing then this file can be used for modifications

WriteValue: write a Proc Format value statement for each variable; this file can be used to
prepare formats

HOW TO GET AND USE THESE PROGRAMS

Overview In order to run this use this program for your project, you need to do the following steps:

1. Create Project Folders

2. Download the Suite Zip File

3. Set Up for the Demonstration

4. Run the Demonstration Program

5. Set Up for Use on Your Library

6. Modifications and Testing

Create Project
Folders

Create the following folders for your FreqLibname project:

contains recommended name

root FreqLibname
sas programs sas
sas data sets sas7b
temporary sas data sets sas7bWork
text files txt

When you are finished your directory structure might look like this:

1 C:\SASprojects
2 C:\SASprojects\FreqLibname
3 C:\SASprojects\FreqLibname\sas
4 C:\SASprojects\FreqLibname\sas7b
5 C:\SASprojects\FreqLibname\sas7bWork
6 C:\SASprojects\FreqLibname\txt

Download the
Suite Zip File

To get the code examples in this paper search www.sascommunity.org for Summarize Mem-
names in Libname.

1. download the .zip file

2. extract files to your project folder for sas programs

Set Up for the
Demonstration

Open the sas programs folder and perform the following steps:

SASv9.cfg customize the configuration file for your project folder

rename file SASv9copy.cfg to SASv9.cfg

edit SASv9.cfg
change the value of SASinitialFolder from
SASinitialFolder ’C:\SASprojects\SmryLibname\sas’
to the name of your project folder:
e.g.: SASinitialFolder ’C:\MyProjects\SmryLibname\sas’

autoexec.sas review the title, filename and libname statements — these are for Windows
— and ensure they conform to your operating system directory specifications

1 * name: autoexec.sas;
2 Title ’SmryLibname: Summary of each Memname in Libname’;
3 Filename Project ’.’ ;
4 Libname Library ’..\sas7b’ ;
5 Libname LibWork ’..\sas7bWork’;

CopySashelpToLibrary.sas submit this program to copy a few SAShelp data sets to the
library

1 *name : Copy-sashelp-to-library.sas;
2 *purpose: provide data sets in Library;
3 * for demonstration and testing;
4 PROC Copy in = sashelp
5 out = Library
6 memtype = data;
7 select CitiDay CitiYr Class;

Run the
Demonstration
Program

Submitting the A0Smry program will create a set of text files for each member in the libref
Library.

• zqCITIDAYreport.txt

• zqCITIYRreport.txt

• zqCLASSreport.txt

Set Up for Use
on Your Library

Make the following changes

autoexec.sas change the directory specification of libname Library to the directory of
the datasets that you wish to report on.

A0Smry.sas change the value of macro variable Path2Txt to the directory where you want
your text reports written; this may be a full directory specification or a (Windows) sibling
folder

Example: %Let Path2Txt = ..\txt\;

Modifications
and Testing

A full set of test files for each routine and subroutine is provided in the suite .zip.

ROUTINES

List of
Programs

This is the list of routines in this section.

Topic Page

CallXProc 9
CallXRpt 11
ProcFreq 11
ProcSmry 13

CallXProc: CALL EXECUTE OF ROUTINES

Overview This is the header record of this program.

CallXProc.sas

1 * name : CallXProc.sas;
2 * description: Call Execute: Procs Freq, Mode, Smry;
3 * purpose : list processing of subroutine;
4

5 * parameters ;
6 * input : ListNames;
7 * process : 1. for each Variable: call procs;
8 * 2. add var Unique to ListNames;
9 * output : 1. from subroutines: ListSmry;

10 * 2.1 sort: out = ListNamesByName;
11 * 2.2 sort: out = ListSmryByName;

This program contains the following steps:

1. Data Structure

2. Make Statement

3. Call Subroutines

4. Link ExecStmnt

5. Add Information

Data Structure Output from this data step is done by call execute so no output data set name is needed.

13 DATA _Null_;
14 attrib Stmnt length = $132
15 Vname length = $ 32;
16 array Mvar(*) $32 Libname MemName Name Type Format;
17

18 retain Testing %eval(0
19 or %sysfunc(getoption(Source2))
20 eq SOURCE2 );

Make
Statement

For each character variable in the array make a global macro variable assignment statement.

21 do until(EndoFile);
22 set &LibWork..ListNames end = EndoFile;
23 %* make macro variable assignment statement:;
24 %* Stmnt = "%let Mvar = value";
25 do I = 1 to dim(Mvar);
26 call vname(Mvar(I) ,Vname);
27 Stmnt = catx(’ ’,’%let ’,Vname,’=’
28 ,Mvar(I) ,’;’);
29 link ExecStmnt;
30 end;

Call
Subroutines

This section calls the various summarization procedures: ProcFreq and, for numerics, Proc-
Mode and ProcSmry.

31 Stmnt = cat(’%Include Project(ProcFreq);’);
32 link ExecStmnt;
33 if Type eq ’N’ then do;
34 Stmnt = cat(’%Include Project(ProcMode);’);
35 link ExecStmnt;
36 Stmnt = cat(’%Include Project(ProcSmry);’);
37 link ExecStmnt;
38 end; %*if Type eq N;
39 end; %*do until(EndoFile);
40 stop;

Link
ExecStmnt

This labeled section enables the program to self report when option source2 is true. It writes
the value of the variable Stmnt to the log. See above for the allocation of the variable Testing
in the data structure.

41 return;
42 ExecStmnt: if Testing then putlog Stmnt=;
43 call execute(cats(’%nrstr(’,Stmnt,’)’));
44 return;
45 run; %*calls executed in this step;

Add
Information

This section adds variable Unique to the report data set ListNames.

47 Data &LibWork..ListNames(drop = Count);
48 do until(Endofile);
49 merge &LibWork..ListSmry
50 (keep = Libname MemName Name Count Level
51 where = (Count = . and Level))
52 &LibWork..ListNames end = EndoFile;
53 by Libname MemName Name;
54 Unique = (NobsData eq Level);
55 if first.Name then output;
56 end;
57 stop;

CallXRpt: CALL EXECUTE REPORTING SUBROUTINE

Overview This is the header record of this program.

1 * name : CallXRpt.sas;
2 * description: Call Execute: Report MemName;
3 * purpose : list processing of subroutine;
4

5 * parameters : ;
6 * input : ListMemNames;
7 * process : 1. for each MemName: call RptMemN;
8 * output : by subroutine;

This program uses the same algorithm as CallXProc to call the subroutine RptMemN.

ProcFreq: SAVE DATA SET

Overview The routine ProcFreq is called by CallXProc. It is a parameterized include file modified
from the FreqAll macro FreqOf. Its parameters are the global macro variables: LibName,
MemName, Name, Type and Format. It calls the subroutine DatStruc.
This program contains the following steps:

1. Internal Parameters

2. Primary Process

3. Standardize Data Structure

4. Read and Output

5. Make Information

6. Output

7. Append

Internal
Parameters

ProcFreq contains two macro variables which determine its output:

1. Nobs2View: the number of rows of the highest and lowest values

2. Order: which controls the proc freq output data set order; note: see also ProcMode.

47 %Let Nobs2View = 3; %* show how many rows?;
48

49 %Let Order = internal;%*default: hi and low values;
50 %* for mode use: replaced by ProcMode;
51 %*Let Order = freq; %*descending count;

Primary
Process

Save the proc freq output data set and rename the variable to the standardized names:
ValuC or ValuN.

53 PROC Freq data = &LibName..&MemName.
54 order = &Order.;
55 format &Name.;%*remove formatting;
56 tables &Name.
57 / list missing noprint
58 out = Freq(rename =
59 (&Name. = Valu&Type.));

Note: the data set Freq is used by the subroutine ProcMode.

Standardize
Data Structure

Call subroutine DatStruc.

61 %Include Project(DatStruc);
62

Read and
Output

Read the data set and output only the lowest and highest rows.

63 do RowNmbr = 1 to NobsFreq;
64 set Freq nobs = NobsFreq
65 point = RowNmbr;
66 %* case 1: output all rows;
67 if NobsFreq le %eval(2 * &Nobs2View. + 2)
68 then link Assigns;
69 else do; %* case 2: lo and hi &Nobs2View. rows;
70 if RowNmbr le &Nobs2View.
71 or RowNmbr ge NobsFreq - &Nobs2View.
72 then link Assigns;
73 else if RowNmbr gt &Nobs2View.
74 then do;
75 RowNmbr = NobsFreq - &Nobs2View.;
76 Level = RowNmbr;
77 end; %*else if RowNmbr gt &Nobs2View.;
78 end; %*else do: case 2;
79 end; %*do RowNmbr;

Note: Compare this single pass algorithm to FreqAll.FreqOf macro.

Make
Information

After the output of the lowest and highest rows make the information row, which contains the
number of rows (Levels) of the proc freq data set and, if the variable is unique, adds a note
saying that the variable is unique: ’is.primekey?’.

81 %* make information row;
82 ValuC = cats(’levels=’,NobsFreq);
83 if NobsData eq NobsFreq then
84 ValuC = cats(ValuC,’:is.primekey?’);
85 Format = ’ ’; ValuF = ’.’; ValuN = . ;
86 Count = . ; Percent = . ; Level = Level -1;
87 link Assigns;
88

Output Add the formatted value.

89 return;
90 Assigns: Level+ +1;
91 if Format ne ’ ’ then do;
92 if Type eq ’C’ then ValuF = putC(ValuC,Format);
93 else ValuF = putN(ValuN,Format);
94 end;
95 output;
96 return;
97 stop;
98 run; %*execute calls here;

Append The freq output is appended to the report data set.

100 PROC Append base = &LibWork..ListSmry
101 data = CommonDataStructure;

ProcSmry: SAVE DATA SET

Overview This subroutine is called by CallXProc.Its parameters are the same as ProcFreq: LibName,
MemName, Name.
This program contains the following steps:

1. Proc Summary

2. Basic Statistics

3. Extra Statistics

4. Transpose

5. Standardize Data Structure

6. Read and Output

7. Append

Proc Summary 3 PROC Summary data = &LibName..&MemName.;
4 var &Name.;
5 output
6 out = Summary
7 ( drop = _Type_ _Freq_)

Basic
Statistics

These are the basic statistics useful in understanding the distribution of a numeric variable.

12 min (&Name.) = min %*;
13 mean (&Name.) = mean %*;
14 median (&Name.) = median %*p50;
15 max (&Name.) = max %*;
16 n (&Name.) = n %*;
17 %*;nmiss (&Name.) = nmiss %*;

Extra Statistics Other statistics may be enabled by adding a semicolon in column 3 which closes the macro
comment and enables the statement. Refer to line 17 for Nmiss, above, for an example.

18 %* p1 (&Name.) = p01 %*;
19 %* p5 (&Name.) = p05 %*;
20 %* p10 (&Name.) = p10 %*;
21 %* p25 (&Name.) = p25 %*q1;
22 %* p50 (&Name.) = p50 %*median;

Transpose The Proc Summary output data set is one row; the Proc Transpose changes the data struc-
ture to one row per statistic.

39 PROC Transpose data = Summary
40 out = SummaryT
41 (keep = Col1 ValuC
42 rename =(Col1 = ValuN ))
43 name = ValuC ;

Standardize
Data Structure

Call subroutine DatStruc.

45 %Include Project(DatStruc);
46

Read Note calculations of percent for N and Nmiss.

47 do until( EndoFile);
48 set SummaryT end = EndoFile;
49 if ValuC in (’n’,’nmiss’) then
50 Percent = 100*(ValuN/NobsData);
51 output;
52 end; %*do until EndoFile;
53 stop;
54 run;

Append 56 PROC Append base = &LibWork..ListSmry
57 data = CommonDataStructure;

SUBROUTINES

List of
Programs

This is the list of subroutines in this section.

Topic Page
DatStruc 15
MakLists 17
ProcMode 19
RptMemN 19

DatStruc: COMMON DATA STRUCTURE

Overview The purpose of DatStruc is to standardize the data structure of each of the procedure out-
puts. This subroutine is called by ProcFreq and ProcSmry.
This program contains the following steps:

1. Output Data Set Name

2. Read Identifiers

3. Set Length of ValuC

4. Allocate Data Structure

5. Initialize Values

6. Self Report When Testing

Output Data
Set Name

Each of the calling routines gets the data set CommonDataStructure.

15 DATA CommonDataStructure
16 (label = ’attrib for ProcFreq and ProcSmry’
17 keep = LibName MemName Name TypeLen
18 ValuC ValuF ValuN
19 Count Percent Level Label);

Read
Identifiers

Read one row from the list processing data set which contains the identifiers and retain all
variables.

20 set &LibWork..ListNames(where = (
21 upcase(LibName) eq "%upcase(&LibName.)"
22 and upcase(MemName) eq "%upcase(&MemName.)"
23 and upcase(Name) eq "%upcase(&Name.)"));
24 retain _all_;

Set Length of
ValuC

Set maximum length of the variable ValuC.

26 %Let LenValuC =
27 %length(levels=123,456,789:is.primekey?);

Allocate Data
Structure

Use the attribute statement to declare the common data structure.

29 attrib TypeLen length = $ %length(C.32767)
30 label = ’Type Len’
31 ValuC length = $ &LenValuC.
32 label = ’Valu C’
33 ValuF length = $ &LenValuC.
34 label = ’formatted’
35 ValuN length = 8 format = best.
36 label = ’Valu N’
37 Count length = 4 format = comma.
38 label = ’N’
39 Percent length = 8 format = 6.2
40 label = ’-%-’
41 Level length = 4
42 Testing length = 4;

Initialize
Values

Note: the Proc Freq output data set supplies either ValuC or ValuN. This retaining ensures
the append works correctly.

43 retain Testing %eval(0
44 or %sysfunc(getoption(Source2))
45 eq SOURCE2 )
46 ValuC ’.’ ValuF ’.’ ValuN .
47 Count . Percent . Level .
48 TypeLen ’?.9’ ;
49 TypeLen = cats(upcase(substr(Type,1,1))
50 ,’.’,Length);

Self Report
When Testing

Conditionally write test messages to log.

51 if Testing then do;
52 put _all_;
53 call execute(’%nrstr(%put _global_;)’);
54 end;

MakLists: MAKE DATA SET FOR LIST PROCESSING

Overview MakLists.sas is called by the main module A0Smry; it prepares the list processing data set
used by both CallXProc and CallXRpt. In FreqAll I used Proc Sql; Phil Mason noted in a
private conversation that Proc Contents is faster. This program has the following steps:

1. Save Proc Contents output

2. Split Proc Contents output

3. Standardize data structure

4. Read data set

5. Recode Contents.Type

6. Assemble Format

7. Output

Save Proc
Contents
Output

The variable Nobs is renamed to differentiate it from NobsFreq, the number of observations
of the Proc Freq data set.

13 PROC Contents data = &Libname.._all_
14 noprint
15 out = &LibWork..ListNames
16 (where = (MemType eq ’DATA’)
17 rename = (Nobs = NobsData) );

Split Proc
Contents
Output

The Proc Contents data set contains more variables than I need so I split it. Note that Nob-
sData is saved in both data sets. ListNames is the list processing data set; ListMemnames
is the first of the final report data sets.

19 DATA &LibWork..ListNames
20 (keep = LibName MemName
21 Name Type Length
22 Label Format
23 Varnum NobsData )
24 &LibWork..ListMemnames
25 (keep = LibName MemName MemLabel
26 NobsData Nvars );

Standardize
Data Structure

The attribute statement declares the order of the variables in the data structure.

27 attrib LibName label = ’LibName’
28 MemName label = ’MemName’
29 Name label = ’Name’
30 Type length = $ 1 label = ’Type’
31 Length label = ’Length’
32 Label label = ’Label’
33 %*$49==sql.dictionary.columns.format length;
34 Format length = $49 label = ’Format’
35 NobsData length = 4 label = ’Nobs Data’
36 Nvars length = 4 label = ’N vars’
37 VarNum label = ’Var Num’ ;
38 retain Nvars 0;

Read Data Set 39 do until(EndoFile);
40 set &LibWork..ListNames
41 (rename = (Type = TypeN))
42 end = EndoFile;
43 by Libname MemName;
44 if first.MemName then Nvars = 0;
45 Nvars+ +1;
46 if last.MemName
47 then output &LibWork..ListMemNames;

Recode
Contents.Type

Recode the Contents.Type numeric variable into a character variable.

48 select(TypeN);%*convert to SQL.Dict.Columns.Type;
49 when(1) Type = ’N’;
50 when(2) Type = ’C’;
51 otherwise;
52 end;

Assemble
Format

Assemble the format from its parts: Format, FormatL and FormatD.

53 if Format ne ’ ’ then do;
54 if FormatL then Format = cats(Format,FormatL
55 ,’.’);
56 else Format = cats(Format,’.’);
57 if FormatD then Format = cats(Format,FormatD);
58 end;
59 output &LibWork..ListNames;
60 end; %* do until(EndoFile);

ProcMode: SORT FREQ DATA SET

Overview This is an optional procedure for numeric variables; it can be disabled in CallXProc.
Variable values are added to the Proc Freq output data set, which is then sorted by de-
scending count. Only the most frequently occurring rows are appended to the summary
report data set.

Add Variable
Values

6 DATA Freq;
7 set CommonDataStructure(obs=1);
8 retain _all_; %*identifiers;
9 retain Level 0 ;

10

11 do until(EndoFile);
12 set Freq end = EndoFile;
13 Level+ +1;
14 ValuC = ’mode’;
15 output;
16 end;
17 stop;

Sort 19 PROC Sort data = Freq;
20 by descending Count;

Append 22 PROC Append base = &LibWork..ListSmry
23 data = Freq
24 (obs = &Nobs2View.);
25 run;

RptMemN: REPORT BY MEMNAME

Overview RptMemN is called by CallXRpt.It writes one summary report for each data set to a text file.
This program contains the following steps:

1. Overview

2. Description

3. Open Output Text File

4. Put Information in Titles

5. Print Attributes

6. Print Summary

7. Close Output

19



Description 1 * name : RptMemN.sas;
2 * description: Report of MemName ;
3 * attributes and summary;
4 * purpose : write summary report to text file;
5

6 * parameters : global: Libname, Memname;
7 * local : ReportName;
8 * input : ListMemNames ListNames ListSmry;
9 * process : get Nobs, Nvars, MemLabel for titles;

10 * print;
11 * output : to text file;

Open Output
Text File

Note: the macro variable Path2Txt is set in main module A0Smry.

13 Proc PrintTo new
14 print = "&Path2Txt.&MemName.-report.txt";

Put
Information in
Titles

Read one row from the list processing data set ListMemNames, which contains the items for
the title statements.

16 %Let NobsData = 0;
17 %Let Nvars = 0;
18 %Let ReportName = Report-Memname;
19 PROC SQL noprint;
20 select NobsData, Nvars, MemLabel
21 into :NobsData, :Nvars, :MemLabel
22 from &LibWork..ListMemnames
23 where Libname eq "%upcase(&Libname.)"
24 and Memname eq "%upcase(&Memname.)";
25 quit;
26 %*note: reassignment == remove leading blanks;
27 %Let NobsData = &NobsData.;
28 %Let Nvars = &Nvars.;
29

30 Title2 "&ReportName.: &Libname..&Memname."
31 " nobs=&NobsData."
32 " nvars=&Nvars." ;
33 Title3 "&ReportName.: MemLabel=%unquote(&MemLabel.)";

Note: MemLabel is unquoted because it may contain either special characters, such as
ampersands or percent signs, or unmatched quotes.

Print Attributes 35 PROC Print data = &LibWork..ListNames
36 (where = (Libname eq "%upcase(&Libname.)"
37 and Memname eq "%upcase(&Memname.)"
38 ) )noobs;
39 label Level = ’Levels’;
40 var VarNum Name Type Length Label
41 Format Level Unique; %*Npos;
42 Title4 "&ReportName.: attributes";

Print Summary 43

44 PROC Print data = &LibWork..ListSmry
45 (where = (Libname eq "%upcase(&Libname.)"
46 and Memname eq "%upcase(&Memname.")
47 ) )label noobs;
48 var ValuC ValuF ValuN Count Percent
49 Level; %* Label;
50 format ValuN;
51 by Name notsorted TypeLen;
52 id Name TypeLen;
53 Title4 "&ReportName.: summary";

Close Output 55 Proc PrintTo;
56 run;

CONCLUSION

Summary FreqAll The data review utility program FreqAll provides a short data set sum-
mary using Proc Freq.

FreqLibname The data review utility suite FreqLibname provides more information,
especially for numerics.

Call Execute This paper shows that call execute is a powerful method for list pro-
cessing.

%Includes Doing list processing with call execute of %Includes can eliminate the
use of macros. This yields clearer code.

Suggested
Reading

Proc DataCheck Abolafia [1, sugi22.229] provides a macro to replace the SUGI Supple-
mental Library Proc DataChk which summarizes numerics.

Fehd: FreqAll Fehd [3, sgf2007.028] wrote the original proc freq code upon which this
paper is based.

Fehd: update After presenting FreqAll in fall 2007 Fehd wrote to the SAS-L listserv to
provide an update: Fehd [2, sasl.225107] which used call execute with a list from
sashelp.vcolumn.

Call Execute and %nrstr Fehd and Carpenter [4, sgf2007.113] demonstrate the timing of
the error of using call execute of macros without the macro function %nrstr.

Acknowledgements My section chair at NESUG, Rob Russell, suggested that I develop FreqAll for every Mem-
Name; Toby Dunn requested formatted values and reviewed early drafts; Peter Flom and
David Cassell requested nvars, Proc Summary median, n and nmiss; Phil Mason noted
that Proc Contents was faster than Proc SQL Dictionary.Columns; and the usual suspects
on SAS-L provided much other encouragement. My colleagues at CDC Susan Katz and
Elizabeth Perez reviewed output and provided suggestions for renaming output text files.

21

http://www2.sas.com/proceedings/sugi22/POSTERS/PAPER229.PDF
http://www2.sas.com/proceedings/forum2007/028-2007.pdf
http://www.listserv.uga.edu/cgi-bin/wa?A2=ind0701A&L=sas-l&P=R14003
http://www2.sas.com/proceedings/forum2007/113-2007.pdf


BIBLIOGRAPHY

[1] Jeffrey M. Abolafia. Proc DataChk revisited: The DataChk macro. In Proceedings of the 22nd Annual SAS Users
Group International Conference, 1997. URL http://www2.sas.com/proceedings/sugi22/POSTERS/
PAPER229.PDF. Posters, 6 pp.; macro to replace SUGI Supplemental Library Proc DataChk; provides summary
of numeric variables.

[2] Ronald Fehd. Re: tip: macro FreqAllVars. In Archives of the SAS-L listserve, 4 Jan. 2007. URL http://
www.listserv.uga.edu/cgi-bin/wa?A2=ind0701A&L=sas-l&P=R14003. Updated algorithm: replace
macro array with call execute.

[3] Ronald J. Fehd. Journeymen’s tools: Data review macro FreqAll — using Proc SQL list processing with
Dictionary.Columns to eliminate macro do loops. In Proceedings of the SAS Global Forum, 2007. URL
http://www2.sas.com/proceedings/forum2007/028-2007.pdf. Coder’s Corner, 10 pp.; attributes,
dictionary.columns, metadata, proc append, proc freq, proc sql, program header; bibliography.

[4] Ronald J. Fehd and Art Carpenter. List processing basics: Creating and using lists of macro variables. In
Proceedings of the SAS Global Forum, 2007. URL http://www2.sas.com/proceedings/forum2007/
113-2007.pdf. Hands On Workshop, 20 pp.; comparison of methods: making and iterating macro arrays,
scanning macro variable, writing calls to macro variable, write to file then include, call execute; 11 examples,
bibliography.

SAS and all other SAS Institute Inc. product or
service names are registered trademarks or trade-
marks of SAS Institute Inc. in the USA and other
countries. R© indicates USA registration.

Author: Ronald Fehd mailto:RJF2@cdc.gov
Centers for Disease Control
4770 Buford Hwy NE
Atlanta GA 30341-3724

To get the code examples in this paper search
www.sascommunity.org for Summarize Memnames
in Libname.

about the author:
education: B.S. Computer Science, U/Hawaii, 1986

SUGI attendee since 1989
SAS-L reader since 1994

experience: programmer: 20+ years
data manager at CDC, using SAS: 18+ years
author: 12+ SUG papers

SAS-L: author: 4,000+ messages to SAS-L since1997
Most Valuable SAS-L contributor: 2001, 2003

Document Production: This paper was
typeset in LATEX. For further information
about using LATEX to write your SUG pa-
per, consult the SAS-L archives:

http://www.listserv.uga.edu/cgi-bin/wa?S1=sas-l
Search for :
The subject is or contains: LaTeX
The author’s address : RJF2
Since : 01 June 2003

22

http://www2.sas.com/proceedings/sugi22/POSTERS/PAPER229.PDF
http://www2.sas.com/proceedings/sugi22/POSTERS/PAPER229.PDF
http://www.listserv.uga.edu/cgi-bin/wa?A2=ind0701A&L=sas-l&P=R14003
http://www.listserv.uga.edu/cgi-bin/wa?A2=ind0701A&L=sas-l&P=R14003
http://www2.sas.com/proceedings/forum2007/028-2007.pdf
http://www2.sas.com/proceedings/forum2007/113-2007.pdf
http://www2.sas.com/proceedings/forum2007/113-2007.pdf
mailto:RJF2@cdc.gov
http://www.sascommunity.org
http://www.sascommunity.org/wiki/Category:Summarize_Memnames_in_Libname
http://www.sascommunity.org/wiki/Category:Summarize_Memnames_in_Libname
http://www.latex-project.org/
http://www.listserv.uga.edu/cgi-bin/wa?S1=sas-l

